ЧАСТЬ VI

КОНСТРУКТЫ ПРОГРАММИРОВАНИЯ

ГЛАВА 14

ПРОЦЕДУРЫ И ПАКЕТЫ

 При строительстве здания надо рассматривать три вещи:

 что оно на правильном месте; что фундамент будет надежен;

 что строительство будет успешно выполнено.

 









Гете

Эта глава обсуждает процедурные возможности ORACLE. Темы этой главы включают обсуждение следующих вопросов:
процедуры, функции и пакеты
процессор PL/SQL в ORACLE
как процедуры, функции и пакеты хранятся в базе данных
как процедуры и функции выполняются ORACLE
Информация этой главы применима лишь к тем системам, которые используют ORACLE с процедурным расширением. Для информации о зависимостях между процедурами, функциями и пакетами и о том, как ORACLE управляет этими зависимостями, обратитесь к главе 16.

[Trusted]

Если вы используете Trusted ORACLE, обратитесь к документу Trusted ORACLE7 Server Administrator's Guide.

Введение в хранимые процедуры и пакеты

ORACLE позволяет вам обращаться к информации в базе данных и манипулировать ей с помощью процедурных объектов схемы, называемых программными единицами PL/SQL. Процедуры, функции и пакеты являются примерами программных единиц PL/SQL.

Хранимые процедуры и функции

Процедура или функция - это объект схемы, который логически группирует множество предложений SQL и других предложений языка программирования PL/SQL для выполнения специфической задачи.  Процедуры и функции создаются в схеме пользователя и сохраняются в базе данных для продолжительного использования. Процедура или функция может исполняться интерактивно через инструмент ORACLE, такой как SQL*DBA, или вызываться явно в коде приложения базы данных, такого как приложение SQL*Forms или программа прекомпилятора, а также может вызываться из другой процедуры или триггера. Рис.14-1 иллюстрирует простую процедуру, хранящуюся в базе данных, и ее вызов из нескольких различных приложений базы данных.

Рис.14-1

Хранимая процедура

 ПРИЛОЖЕНИЯ

        БАЗЫ ДАННЫХ

                 -------------------¬

                 ¦  Программный код ¦

                 ¦  .               ¦

                 ¦  .               ¦

-----------------+-¬Программный код ¦

¦  Программный код ¦.               ¦              HIRE_EMP(...)

¦  .               ¦HIRE_EMP(...); -+----------
   --------------¬

¦  .               ¦.               ¦    ------
   ¦ BEGIN       ¦

¦  Программный код ¦Программный код ¦    ¦  ---
   ¦ .           ¦

¦  .               +-----------------    ¦  ¦      ¦ .           ¦

¦  HIRE_EMP(...); -+----------------------  ¦      ¦ END;        ¦

¦  .          -----+-------------¬          ¦      +-------------+

¦  Программный¦  Программный код ¦          ¦

L-------------¦  .               ¦          ¦      L - -- T - - --

              ¦  .               ¦          ¦    г========¦==========¬

              ¦  Программный код ¦          ¦    ¦    База данных    ¦

              ¦  .               ¦          ¦    ¦                   ¦

              ¦  HIRE_EMP(...); -+-----------    L===================-

              ¦  .               ¦

              ¦  Программный код ¦

              L-------------------

Сама хранимая процедура на рис.14-1, которая вставляет запись о сотруднике в таблицу EMP, показана на рис.14-2.

Рис.14-2

Процедура HIRE_EMP

            PROCEDURE hire_emp (name VARCHAR2, job VARCHAR2,

                                mgr NUMBER, hiredate DATE, sal NUMBER,

                                comm NUMBER, deptno NUMBER);

          -----------------------------------------------¬

          ¦ BEGIN                                        ¦

          ¦  .                                           ¦

          ¦  .                                           ¦

          ¦     INSERT INTO emp VALUES                   ¦

          ¦      (emp_sequence.NEXTVAL, name, job, mgr,  ¦

          ¦       hiredate, sal, comm, deptno);          ¦

          ¦  .                                           ¦

          ¦  .                                           ¦

          ¦ END;                                         ¦

          L-----------------------------------------------

Любое из приложений, показанных на рис.14-1, может вызывать процедуру HIRE_EMP. Альтернативно, привилегированный пользователь может использовать SQL*DBA для выполнения процедуры HIRE_EMP с помощью следующей команды:

EXECUTE hire_emp ('TSMITH', 'CLERK', 1037, SYSDATE, \ 500, NULL, 20);

В результате этого предложения EXECUTE в таблицу EMP будет вставлена новая запись для сотрудника TSMITH (а также будут выполнены другие вычисления и предложения SQL, которые могли содержаться в процедуре).

Пакеты

Пакет предоставляет способ группирования связанных процедур и функций, курсоров и переменных в общую единицу и сохранение этой единицы в базе данных для продолжительного использования.  Аналогично независимым процедурам и функциям, пакетированные процедуры и функции могут явно вызываться приложениями и пользователями. Рис.14-3 иллюстрирует пакет, в котором инкапсулирован ряд процедур, используемых для управления базой данных персонала.

Рис.14-3

Хранимый пакет

Приложения базы данных явно вызывают пакетированные процедуры.  Получив привилегии для пакета EMP_MGMT, пользователь может явно выполнять любую из процедур в этом пакете. Например, следующее предложение можно было бы выдать в SQL*DBA, чтобы выполнить пакетированную процедуру HIRE_EMP:

EXECUTE emp_mgmt.hire_emp ('TSMITH', 'CLERK', 1037, SYSDATE, \ 500, NULL, 20);

Пакеты обладают несколькими преимуществами в разработке и производительности по сравнению с независимыми хранимыми процедурами и функциями, как описано в секции «Пакеты» на странице 14-11.

PL/SQL

PL/SQL - это процедурное языковое расширение SQL, принадлежащее фирме Oracle. PL/SQL предоставляет возможность определять и выполнять программные единицы PL/SQL, такие как процедуры, функции и пакеты. Определение, компиляция и исполнение программных единиц PL/SQL управляется ПРОЦЕССОРОМ PL/SQL.  Процессор PL/SQL представляет собой специальную компоненту многих продуктов Oracle, в том числе сервера ORACLE, и предназначен для обработки PL/SQL. Рис.14-4 иллюстрирует процессор PL/SQL, содержащийся в сервере ORACLE.

Рис.14-4

Процессор PL/SQL и ORACLE

                                          Сервер ORACLE

                         ----------------------------------------------¬

                         ¦                                             ¦

                         ¦     SGA                  Процессор PL/SQL   ¦

    Приложение           ¦  -------------------¬   ------------------¬ ¦

    базы данных          ¦  ¦                  ¦   ¦ --------------¬ ¦ ¦

  -------------------¬   ¦  ¦ Процедура        ¦   ¦ ¦ Исполнитель ¦ ¦ ¦

  ¦                  ¦   ¦  ¦ ---------------¬ ¦ --
 ¦ процедурных ¦ ¦ ¦

  ¦ Программный код  ¦   ¦  ¦ ¦BEGIN         ¦ ¦ ¦ ¦ ¦ предложений ¦ ¦ ¦

  ¦ Программный код  ¦   ¦  ¦ ¦  процедурное ¦ ¦ ¦ ¦ L------T------- ¦ ¦

  ¦ Программный код  ¦   ¦  ¦ ¦  процедурное ¦ ¦ ¦ L--------+--------- ¦

  ¦ Вызов процедуры -+---+--+
¦  SQL         +-+--          ¦          ¦

  ¦ Программный код  ¦   ¦  ¦ ¦  процедурное ¦ ¦            ¦SQL       ¦

  ¦ Программный код  ¦   ¦  ¦ ¦  SQL         ¦ ¦            ¦          ¦

  ¦                  ¦   ¦  ¦ ¦END;          ¦ ¦     -------​------¬   ¦

  L-------------------   ¦  ¦ +--------------+ ¦     ¦ Исполнитель ¦   ¦

                         ¦  L-+--------------+--     ¦ предложений ¦   ¦

                         ¦    ¦              ¦       ¦     SQL     ¦   ¦

                         ¦                           L--------------   ¦

                         L----+--------------+--------------------------

                              L- - - T- - - --

                         г===========¦============¬

                         ¦      База данных       ¦

                         ¦                        ¦

                         L========================-

Заметьте, что процедура (или пакет) хранится в базе данных.  Когда процедура, хранящаяся в базе данных, вызывается из приложения, откомпилированная процедура (или весь пакет) загружается в разделяемый пул в области SGA, и исполнитель PL/SQL вместе с исполнителем предложений SQL обрабатывают предложения в этой процедуре.

Следующие продукты Oracle также включают в себя процессор PL/SQL:
SQL*Forms версии 3

SQL*Menu версии 5

· SQL*ReportWriter версии 2

Хотя многие продукты Oracle имеют PL/SQL как свою компоненту, эта глава посвящена тем процедурам и пакетам, которые могут быть сохранены в базе данных ORACLE, и обрабатываются процессором PL/SQL сервера ORACLE. Возможности PL/SQL каждого инструмента ORACLE описываются в соответствующем руководстве по инструменту ORACLE.
Языковые конструкты для процедур и пакетов

Процедуры, функции и пакеты могут включать следующие языковые конструкты PL/SQL:
переменные и константы

курсоры

· исключения

Ниже дается краткое описание этих конструктов; для более подробной информации обратитесь к документу PL/SQL User's Guide and Reference.
Переменные и константы

Переменные и константы могут объявляться в процедуре, функции или пакете. Переменная или константа может использоваться в предложении SQL или PL/SQL для приема или передачи значения, где оно необходимо.

Замечание: Некоторые интерактивные инструменты, такие как SQL*DBA, позволяют вам определять переменные в текущей сессии.  Переменные, объявленнные этим способом, могут использоваться так же, как и переменные, объявляемые в процедурах и пакетах.

Курсоры

Курсоры могут явно объявляться в процедуре, функции или пакете для облегчения ориентированной на записи обработки данных ORACLE. Курсоры могут также неявно объявляться (для поддержки других действий манипулирования данными) самим процессором PL/SQL.

Исключения

PL/SQL позволяет вам явно обрабатывать как внутренние, так и определенные пользователем условия ошибок, называемых ИСКЛЮЧЕНИЯМИ, которые могут возникать в ходе обработки кода PL/SQL. Внутренние исключения вызываются незаконными операциями, такими как деление на 0, или ошибками ORACLE, возвращаемыми коду PL/SQL. Пользовательские исключения определяются явно, и возбуждаются внутри блока PL/SQL для управления обработкой ошибок, специфичной для приложения (например, отрицательный баланс при дебитовании счета).

Когда исключение ВОЗБУЖДАЕТСЯ, нормальное выполнение кода PL/SQL останавливается, и вызывается программа, называемая обработчиком исключений. Для любого внутреннего или пользовательского исключения может быть написан специфичный обработчик исключений.

Процедуры и функции

ORACLE может обрабатывать как индивидуальные предложения SQL, так и процедуры и функции. ПРОЦЕДУРА или ФУНКЦИЯ - это объект схемы, который состоит из сгруппированных вместе предложений и других конструктов PL/SQL, хранится в базе данных и выполняется как единица для решения специфической проблемы или выполнения множества связанных задач. Процедуры и функции позволяют вам комбинировать легкость и гибкость SQL с процедурными возможностями структурного языка программирования.

Например, следующее предложение создает процедуру CREDIT_ACCOUNT, которая кредитует банковский счет.

CREATE PROCEDURE credit_account

(acct NUMBER, credit NUMBER) AS

/* Эта процедура принимает два аргумента: номер счета и денежную сумму, которой кредитуется этот счет.
Если указанный счет не существует, создается новый счет. */

            old_balance  NUMBER;

            new_balance  NUMBER;

          BEGIN

              SELECT balance INTO old_balance FROM accounts

                  WHERE acct_id = acct

                  FOR UPDATE OF balance;

              new_balance := old_balance + credit;

              UPDATE accounts SET balance = new_balance

                  WHERE acct_id = acct;

              COMMIT;

          EXCEPTION

              WHEN NO_DATA_FOUND THEN

                INSERT INTO accounts (acct_id, balance)

                  VALUES (acct, credit);

              WHEN OTHERS THEN

                ROLLBACK;

        END credit_account;

Заметьте, что в процедуру CREDIT_ACCOUNT включены как предложения SQL, так и предложения PL/SQL.

Процедуры и функции почти идентичны. Единственными различиями между ними являются то, что функция всегда возвращает единственное значение вызывающей программе, а процедуры не возвращают значений, и то, что функции не допускают аргументов INOUT и OUT. Для простоты ниже в этой главе всюду употребляется термин «процедура» вместо «процедура или функция», если специально не оговорено иное.

Как используются процедуры

При проектировании и использовании любых хранимых процедур следует придерживаться следующих принципов:
Определяйте процедуру для выполнения единственной, сфокусированной задачи. Не определяйте длинных процедур с несколькими различными подзадачами; в противном случае подзадачи, общие для многих процедур, будут без всякой необходимости дублироваться в кодах разных процедур.
Не определяйте процедур, которые повторяют возможности, уже предоставляемые другими средствами ORACLE. Например, не определяйте процедур для того, чтобы вводить в действие простые правила целостности данных, которые вы можете легко реализовать через декларативные ограничения целостности.
Приложения для процедур

Процедуры предоставляют преимущества в следующих областях:
Защите

Производительности

распределении памяти

продуктивности

· целостности

Защита

Хранимые процедуры повышают безопасность данных. Ваш АБД может ограничить доступ пользователей к определенным операциям с базой данных, предоставляя доступ лишь через процедуры и функции.
Например, АБД может предоставить пользователям право выполнения хранимой процедуры, обновляющей таблицу, но не давать доступа к самой таблице. Когда пользователь вызывает процедуру, эта процедура выполняется с привилегиями владельца процедуры. Так как пользователи имеют лишь привилегию выполнять процедуру, но не имеют привилегий опрашивать, обновлять или удалять данные из соответствующих таблиц, возможности пользователей по манипулированию данными ограничиваются тем, что заложено в процедуру.

Производительность

Хранимые процедуры могут улучшить производительность базы данных. Использование процедур существенно сокращает объем информации, пересылаемой по сети, по сравнению с выдачей индивидуальных предложений SQL или пересылкой в ORACLE текста целого блока PL/SQL. Более того, так как откомпилированная форма процедуры хранится в базе данных готовой к выполнению, для исполнения ее кода не требуется шага компиляции. Кроме того, если процедура уже присутствует в разделяемом пуле в SGA, то не требуется ее извлечения с диска, и выполнение может начаться немедленно.

Распределение памяти

Хранимые процедуры используют преимущества разделяемой памяти ORACLE. Так, лишь одна копия процедуры должна быть загружена в память, чтобы быть доступной всем пользователям. Использование одного и того же кода совместно многими пользователями приводит к существенной экономии памяти, требуемой приложениям.

Продуктивность

Хранимые процедуры повышают продуктивность разработки.  Проектируя приложения вокруг общей библиотеки хранимых процедур, вы можете избежать неоднократного кодирования повторяющихся операций.

Предположим, например, что несколько различных приложений вызывают хранимую процедуру, которая управляет вставками, обновлениями и удалениями в таблице EMP. Если метод управления изменится, потребуется пересмотреть лишь одну процедуру, а не все приложения.

Целостность

Хранимые процедуры улучшают целостность и согласованность ваших приложений. Создавая приложения вокруг библиотеки хранимых подпрограмм, вы снижаете вероятность ошибок кодирования.

Например, процедуру или функцию требуется протестировать лишь однократно, чтобы гарантировать, что она возвращает точный результат. После проверки, процедура может использоваться в любом числе приложений, поставляя им гарантированно достоверные результаты. При любом изменении структур данных, к которым обращается процедура, перекомпилировать потребуется лишь эту процедуру; приложения, вызывающие эту процедуру, не потребуют никаких модификаций (даже перекомпиляции).

Сравнение анонимных блоков и хранимых процедур

Анонимный блок PL/SQL создается посылкой в сервер ORACLE непоименованного блока PL/SQL из инструмента Oracle или приложения. ORACLE компилирует этот блок PL/SQL и помещает его откомпилированную версию в разделяемый пул в SGA, но не сохраняет ни исходный текст блока, ни его откомпилированную версию в базе данных для последующего использования.

Разделяемый SQL позволяет повторно (и совместно) использовать анонимный блок PL/SQL, уже находящийся в разделяемом пуле, до тех пор, пока он не будет вытеснен из разделяемого пула.

Альтернативно, разделяемая процедура создается и сохраняется в базе данных как объект. Однажды созданный и откомпилированный, этот поименованный объект может исполняться без перекомпиляции.  Кроме того, в словаре данных запоминается информация о зависимостях, благодаря которой гарантируется действительность каждой хранимой процедуры.

В общем, вынеся блоки PL/SQL из ваших приложений и поместив их в хранимые процедуры базы данных, вы избежите излишних перекомпиляций во время выполнения и улучшите общую производительность приложений и ORACLE.

Сравнение независимых и пакетированных процедур

Хранимые процедуры, определенные вне контекста пакета, называются НЕЗАВИСИМЫМИ ПРОЦЕДУРАМИ. Процедуры, определенные внутри пакета (пакетированные), рассматриваются как часть пакета. См. секцию «Пакеты» на странице 14-11 для информации о преимуществах пакетов.

Отслеживание зависимостей для хранимых процедур

Хранимая процедура зависит от объектов, к которым есть обращения в ее теле. ORACLE автоматически отслеживает и управляет такими зависимостями. Например, если вы изменяете определение таблицы, к которой обращается процедура, то эта процедура должна быть перекомпилирована, чтобы удостовериться, что она работает как положено; во многих случаях ORACLE автоматически администрирует такое управление зависимостями. См. главу 16 для дополнительной информации об отслеживании зависимостей.

Пакеты

Пакеты предоставляют способ инкапсулирования связанных процедур, функций и ассоциированных курсоров и переменных в общую единицу и сохранение этой единицы в базе данных.

Например, следующие два предложения создают спецификацию и тело пакета, который содержит несколько процедур и функций для обработки банковских транзакций.

 CREATE PACKAGE bank_transactions AS

            minimum_balance  CONSTANT NUMBER := 100.00;

            PROCEDURE apply_transactions;

            PROCEDURE enter_transaction (acct    NUMBER,

                                         kind    CHAR,

                                         amount  NUMBER);

        END bank_transactions;

        CREATE PACKAGE BODY bank_transactions AS

 /* Пакет для ввода банковских транзакций */

 new_status CHAR(20); /* Глобальная переменная для записи

 состояния применяемой транзакции.

 Используется для обновления в

 процедуре APPLY_TRANSACTIONS. */

 PROCEDURE do_journal_entry (acct NUMBER,

 kind CHAR) IS

 /* Регистрирует в журнале каждую банковскую транзакцию,

 применяемую процедурой APPLY_TRANSACTIONS. */

 BEGIN

                INSERT INTO journal

                    VALUES (acct, kind, sysdate);

                IF kind = 'D' THEN

                    new_status := 'Debit applied';

                ELSIF kind = 'C' THEN

                    new_status := 'Credit applied';

                END IF;

            END do_journal_entry;

 PROCEDURE credit_account (acct NUMBER, credit NUMBER) IS

 /* Кредитует счет указанной суммой. Если счет не существует,

 процедура сначала создает новый счет. */

                old_balance  NUMBER;

                new_balance  NUMBER;

            BEGIN

                SELECT balance INTO old_balance FROM accounts

                    WHERE acct_id = acct

                    FOR UPDATE OF balance;  -- блокирует для обновления

                new_balance := old_balance + credit;

                UPDATE accounts SET balance = new_balance

                    WHERE acct_id = acct;

                do_journal_entry(acct, 'C');

            EXCEPTION

                WHEN NO_DATA_FOUND THEN /* Создать новый счет */

                  INSERT INTO accounts (acct_id, balance)

                    VALUES (acct, credit);

                do_journal_entry(acct, 'N');

                WHEN OTHERS THEN /* Вернуть прочие ошибки приложению */

                    new_status := 'Error: ' || SQLERRM(SQLCODE);

            END credit_account;

 PROCEDURE debit_account (acct NUMBER, debit NUMBER) IS

            /* Дебитует счет, если номер счета действителен,

             * и на счете достаточная сумма. */

                old_balance         NUMBER;

                new_balance         NUMBER;

                insufficient_funds  EXCEPTION;

            BEGIN

                SELECT balance INTO old_balance FROM accounts

                    WHERE acct_id = acct

                    FOR UPDATE OF balance;

                new_balance := old_balance - debit;

                IF new_balance >= minimum_balance THEN

                    UPDATE accounts SET balance = new_balance

                        WHERE acct_id = acct;

                    do_journal_entry(acct, 'D');

                ELSE

                    RAISE insufficient_funds;

                END IF;

 EXCEPTION

                WHEN NO_DATA_FOUND THEN

                    new_status := 'Nonexistent account';

                WHEN insufficient_funds THEN

                    new_status := 'Insufficient funds';

                WHEN OTHERS THEN /* Вернуть прочие ошибки приложению */

                    new_status := 'Error: ' || SQLERRM(SQLCODE);

            END debit_account;

            PROCEDURE apply_transactions IS

 /* Применяет транзакции, ожидающие в таблице транзакций,

 к банковским счетам. Используется через регулярные

 интервалы для обновления банковских счетов, не мешая

 вводу новых транзакций. */

 CURSOR trans_cursor IS

                    SELECT acct_id, kind, amount FROM transactions

                        WHERE status = 'Pending'

                        ORDER BY time_tag

                        FOR UPDATE OF status;

            BEGIN

                FOR trans IN trans_cursor LOOP /* неявное извлечение */

                    IF trans.kind = 'D' THEN

                        debit_account(trans.acct_id, trans.amount);

                    ELSIF trans.kind = 'C' THEN

                        crebit_account(trans.acct_id, trans.amount);

                    ELSE

                        new_status := 'Rejected';

                    END IF;

                    /* Обновить таблицу транзакций, чтобы возвратить

                       результат применения данной транзакции. */

                    UPDATE transactions SET status = new_status

                        WHERE CURRENT OF trans_cursor;

                END LOOP;

                COMMIT; /* Освободить блокировки в таблице транзакций */

            END apply_transactions;

 PROCEDURE enter_transaction (acct    NUMBER,

                                         kind    CHAR,

                                         amount  NUMBER) IS

 /* Вводит банковскую транзакцию в таблицу транзакций.

 Новая транзакция всегда помещается в эту "очередь"

 перед тем, как она будет применена к указанному счету

 процедурой APPLY_TRANSACTIONS. Поэтому можно вводить

 одновременно много транзакций, не мешая друг другу. */

 BEGIN

                INSERT INTO transactions

                    VALUES (acct, kind, amount, 'Pending', SYSDATE);

                COMMIT;

            END enter_transaction;

        END bank_transactions;

Помогая администратору базы данных или разработчику приложения организовывать такие программы в управляемые единицы, пакеты в то же время предлагают расширение возможностей и улучшение производительности базы данных.

Применения пакетов

Пакеты используются для определения взаимосвязанных процедур, переменных и курсоров. С помощью пакетов часто достигаются преимущества в следующих областях:
инкапсуляция связанных процедур и переменных

разделение спецификации пакета и тела этого пакета

объявление общих и личных процедур, переменных, констант и курсоров
определение переменных, сохраняющих значение между вызовами
улучшение производительности
Инкапсуляция

Пакеты позволяют вам инкапсулировать логически связанные типы данных, хранимые процедуры, переменные и т.д. в единую поименованную единицу, хранящуюся в базе данных. Каждый пакет легко понять, а интерфейсы между пакетами просты, ясны и хорошо определены. Это облегчает разработку приложений.

Инкапсуляция процедурных конструктов в пакет также облегчает управление привилегиями. Назначение привилегии на использование пакета делает все конструкты в этом пакете доступными тому, кто получает привилегию.

Общие и личные данные и процедуры

С помощью пакетов вы можете указывать, какие типы, объекты и подпрограммы являются:
общими 
Непосредственно видимыми пользователю пакета

личными 
Скрытыми от пользователя пакета

Например, пакет может содержать десять процедур. Однако этот пакет может быть определен так, что лишь три процедуры являются общими, и потому доступными для выполнения пользователю пакета; остальные процедуры являются личными, и могут использоваться лишь процедурами внутри самого пакета.

Не путайте общие и личные переменные пакета с грантами PUBLIC, которые описываются в главе 17.

Разделение спецификации и тела пакета

Пакет создается из двух частей: спецификации и тела.  СПЕЦИФИКАЦИЯ пакета объявляет все общие конструкты этого пакета, а тело определяет все конструкты (как общие, так и личные). Это разделение на две части предоставляет следующие преимущества:
Определяя спецификацию пакета отдельно от тела пакета, разработчик получает большую гибкость в цикле разработки.  После создания спецификации к общим процедурам можно обращаться еще до того, как создано тело пакета.
Тела процедур, содержащиеся в теле пакета, можно изменять независимо от из публично объявленных спецификаций в спецификации пакета. Пока не изменяется сама спецификация процедуры, объекты, которые обращаются к изменяющимся процедурам, никогда не помечаются как недействительные; иными словами, они никогда не требуют перекомпиляции. Для дополнительной информации о зависимостях обратитесь к главе 16.
Улучшение производительности

Использование пакетов вместо независимых хранимых процедур дает следующие преимущества:
Когда вы вызываете пакетированную подпрограмму первый раз, в память загружается весь пакет. Эта загрузка выполняется за одну операцию, в отличие от отдельных загрузок, требуемых для независимых процедур. Поэтому последующие вызовы других подпрограмм этого пакета не требуют операций ввода-вывода.
Тело пакета можно заменять и перекомпилировать, не затрагивая его спецификацию. Как следствие, объекты, которые обращаются к конструктам пакета (всегда через его спецификацию), никогда не требуют перекомпиляции, пока не будет изменена также спецификация пакета. Благодаря пакетам, количество перекомпиляций может быть минимизировано, что улучшает общую производительность базы данных.
Отслеживание зависимостей для пакетов

Пакет зависит от объектов, к которым обращаются процедуры и функции, определенные в его теле. ORACLE автоматически отслеживает и управляет такими зависимостями. Подробнее об этом см. в главе 16.

Как ORACLE хранит процедуры и пакеты

Когда вы создаете процедуру или пакет, ORACLE автоматически выполняет следующие шаги:
1.
Компилирует процедуру или пакет.

2.
Сохраняет откомпилированный код в основной памяти.

3.
Сохраняет процедуру или пакет в базе данных.

Компиляция процедур и пакетов

Компиляцию исходного кода осуществляет компилятор PL/SQL. Этот компилятор является частью процессора PL/SQL, содержащегося в ORACLE. Если по время компиляции происходит ошибка, то возвращается сообщение об ошибке. Информация о том, как идентифицировать ошибки компилятора, содержится в документе ORACLE7 Server Application Developer's Handbook.

Сохранение откомпилированного кода в памяти

ORACLE кэширует откомпилированную процедуру или пакет в разделяемом пуле в области SGA. Это позволяет быстро исполнять откомпилированный код и использовать его совместно всеми пользователями. Откомпилированная версия процедуры или пакета остается в разделяемом пуле столько, сколько это позволит алгоритм LRU, даже если первоначальный создатель процедуры закончит свою сессию. См. секцию «Разделяемый пул» на странице 9-17 для более подробной информации о буфере разделяемого пула.

Сохранение процедур и пакетов в базе данных

Во время создания и компиляции процедуры или пакета ORACLE автоматически сохраняет в базе данных следующую информацию:

имя объекта 
ORACLE использует это имя для идентификации процедуры или пакета. Вы задаете это имя в предложениях CREATE PROCEDURE, CREATE FUNCTION, CREATE PACKAGE и CREATE PACKAGE BODY.

исходный код 
Компилятор PL/SQL разбирает исходный код и генерирует 
и дерево 
разобранное представление исходного кода, называемое ДЕРЕВОМ

разбора 
РАЗБОРА.

псевдокод 
Компилятор PL/SQL генерирует ПСЕВДОКОД, или P-код, на основе

(P-код) 
разобранного кода. Процессор PL/SQL исполняет этот код при вызове процедуры или пакета.

сообщения 
ORACLE может генерировать ошибки во время компиляции процедуры

об ошибках 
или пакета.

Чтобы избежать излишних перекомпиляций процедуры или пакета, в базе данных сохраняются как дерево разбора, так и псевдокод объекта. Это позволяет процессору PL/SQL при обращении к процедуре или пакету прочитать откомпилированную версию в разделяемый пул в SGA, если его еще нет в SGA. Дерево разбора используется при компиляции кода, вызывающего процедуру.

Все части процедур базы данных сохраняются в табличном пространстве SYSTEM (словаре данных) соответствующей базы данных. Администратор базы данных должен планировать размер табличного пространства SYSTEM, имея в виду, что все хранимые процедуры требуют места в этом табличном пространстве.

Как ORACLE исполняет процедуры и пакеты

Когда вызывается независимая или пакетированная процедура, ORACLE выполняет следующие шаги:
1.
Проверяет права пользователя.

2.
Проверяет действительность процедуры.

3.
Исполняет процедуру.

Проверка прав пользователя

ORACLE проверяет, что вызывающий пользователь владеет вызываемой процедурой или имеет привилегию EXECUTE для процедуры или объемлющего ее пакета. Пользователю, вызывающему процедуру, не требуется привилегий доступа к процедурам или объектам, к которым есть обращения в данной процедуре; такие привилегии требуются лишь для владельца процедуры или пакета.

Проверка действительности процедуры

ORACLE проверяет состояние процедуры или пакета в словаре данных на действительность. Объект (процедура или пакет) помечается как НЕДЕЙСТВИТЕЛЬНЫЙ, если после последней компиляции этого объекта произошло одно из следующих событий:
Хотя бы один из объектов (таблиц, обзоров, других процедур), к которым обращается процедура, был изменен (ALTER) или удален (DROP). (Например, вы добавили в таблицу столбец.)
Системная привилегия была отобрана у PUBLIC или у владельца процедуры или пакета.

Привилегия объекта для одного или нескольких объектов, к которым обращается процедура или пакет, была отобрана у PUBLIC или у владельца процедуры или пакета.

Процедура ДЕЙСТВИТЕЛЬНА, если она не была отмечена как недействительная по одной из перечисленных выше причин.

Если вызывается действительная независимая или пакетированная процедура, то исполняется ее откомпилированный код.

Если вызывается недействительная независимая или пакетированная процедура, то она автоматически перекомпилируется перед исполнением.

Для полного обсуждения действительных и недействительных процедур и пакетов, а также подробного обсуждения вопросов зависимостей, обратитесь к главе 16.

Исполнение процедуры

Процессор PL/SQL исполняет процедуру или пакет различными методами, в зависимости от ситуации:
Если процедура действительна и сейчас находится в памяти, то процессор PL/SQL просто исполняет ее псевдокод.
Если процедура действительна, но сейчас не находится в памяти, то процессор PL/SQL загружает с диска в память ее откомпилированный псевдокод и исполняет его. В случае пакета, загружаются все конструкты пакета (т.е. все процедуры, переменные и т.п., которые были откомпилированы как единица кода).

Процессор PL/SQL обрабатывает процедуру предложение за предложением, исполняя все процедурные предложения самостоятельно и передавая предложения SQL исполнителю предложений SQL, как показано на рис.14-3 на странице 14-7.

